

Mathematical models in population dynamics

The per capita growth rate, r , is defined by

$$r = \frac{\text{birth rate} - \text{death rate}}{\text{population size}}$$

The maximum value that r can have for an organism is r_{max} , called the biotic potential. They range from about 0.03 per year for large mammals to about 10 per year for insects and about 10,000 per year for bacteria.

Let us denote by

$x(t)$ - the size of the population at the moment $t > 0$

x_0 - the size of the population at the initial moment $t_0 = 0$

The variation of the population size is given by

$$\frac{x(t + \Delta t) - x(t)}{\Delta t}$$

the growth rate is

$$\lim_{\Delta t \rightarrow 0} \frac{x(t + \Delta t) - x(t)}{\Delta t} = \frac{d}{dt} x(t)$$

The per capita growth rate, r , is

$$r = \frac{\frac{d}{dt} x(t)}{x(t)}$$

```
> with(DEtools):
```

```
> with(plots):
```

```
Warning, the name changecoords has been redefined
```

The Exponential Model for Growth

Thomas Malthus in 1798, give the first model predicting population size assuming that the per capita growth rate is constant

$$\frac{\frac{d}{dt} x(t)}{x(t)} = r = \text{constant}$$

thus the Malthus model is

$$\begin{aligned} \frac{d}{dt} x(t) &= r x(t) \\ x(0) &= x_0 \end{aligned}$$

```
> M_eq:=diff(x(t),t)=r*x(t);
```

$$M_eq := \frac{d}{dt} x(t) = r x(t)$$


```
> ans:=dsolve({M_eq,x(0)=x0},x(t));
```

$$ans := x(t) = x0 e^{(rt)}$$

```
> x_sol:=unapply(rhs(ans),t,r,x0);
```

$$x_sol := (t, r, x0) \rightarrow x0 e^{(rt)}$$

```
> plot([x_sol(t,-  
0.5,100),x_sol(t,0,100),x_sol(t,0.5,100)],t=0..2,color=[red,blue,black]);
```



The per capita growth rate parameter r is often given in terms of doubling time. Denote by $T2$ the time when the population size reaches twice its initial value.

```
> eq:=x_sol(T2,r,x0)=2*x0;
```

$$eq := x0 e^{(rT2)} = 2 x0$$

```
> solve(eq,T2);
```

$$\frac{\ln(2)}{r}$$

```
> evalf(%);
```

$$\frac{0.6931471806}{r}$$

```
>
```

$$T2 = \frac{\ln(2)}{r} \approx \frac{.7}{r} \text{ or } r = \frac{\ln(2)}{T2} \approx \frac{.7}{T2}$$

Growth parameters can be determined from experimental data.

U.S. population census. (Source: Statistical Abstracts of the United States, 113th ed., Bureau of the Census, U.S. Department of Commerce, Washington, DC, 1993.)

1790: 3929214	1860: 31433321	1930: 122775046
1800: 5308483	1870: 39818449	1940: 131669275
1810: 7239881	1880: 50155783	1950: 151325798
1820: 9638453	1890: 62947714	1960: 179323175
1830: 12866020	1900: 75994575	1970: 203302031
1840: 17069453	1910: 91972266	1980: 226545805
1850: 23191876	1920: 105710620	1990: 248709873

```
> SUA_pop:=[3.92, 5.3, 7.23,9.63, 12.86,17.06, 23.19, 31.43,  
39.81,50.15, 62.94,75.99,91.97, 105.71, 122.77, 131.66, 151.32,  
179.32, 203.3, 226.54, 248.7];
```

```
SUA_pop := [3.92, 5.3, 7.23, 9.63, 12.86, 17.06, 23.19, 31.43, 39.81, 50.15, 62.94, 75.99,  
91.97, 105.71, 122.77, 131.66, 151.32, 179.32, 203.3, 226.54, 248.7]
```


```
> SUA_pop[21];
```

248.7

```
> N:=21;
```

N:=21

```
> g1:=plot([[n-  
1,SUA_pop[n]]$n=1..N],style=point,symbol=circle,color=blue):  
> display(g1);
```



```
> x0:=SUA_pop[1];x1:=SUA_pop[2];
```

$x_0 := 3.92$

$x_1 := 5.3$

```
> eq:=x_sol(1,r,x0)=x1;
```


$eq := 3.92 e^r = 5.3$

```
> r1:=solve(eq,r);
```

$r_1 := 0.3016151668$

```
> g2:=plot(x_sol(t,r1,x0),t=0..20, color=red):
```

```
> display(g1,g2);
```



>

Exponential growth entails two parameters, initial population size x_0 and growth rate r . Given n experimental data values, $(t_0, x_0), (t_1, x_1), (t_2, x_2), \dots, (t_n, x_n)$, we would like to find the specific parameter values for the experiment. This can be done by the method of least squares. We first put the equation into a form linear with respect to the parameters

$$\ln x = \ln x_0 + r t$$

The squared error is then given by

$$E(r) = \sum_{i=1}^n (\ln(x_i) - (\ln(x_0) + r t_i))^2$$

$$E(r) \rightarrow \min \text{ then } E'(r) = 0$$

$$2 \left(\sum_{i=1}^n (\ln(x_i) - (\ln(x_0) + r t_i)) (-t_i) \right) = 0$$


Solving this equation with respect to parameter r we get

$$r = \frac{\sum_{i=1}^n t_i (\ln(x_i) - \ln(x_0))}{\sum_{i=1}^n t_i^2}$$

```
> r2:=sum(i*(ln(SUA_pop[i])-ln(x0)),i=1..N)/sum(i^2,i=1..N);
      r2 := 0.2209685078
```

```
> g3:=plot(x_sol(t,r2,x0),t=0..20, color=red):
```

```
> display(g1,g3);
```



>

The logistic model (Verhulst)

Pierre Verhulst in 1845 considered that per capita growth rate is not constant, he assumed that depends on the population size at that moment.

$$\frac{d}{dt}x(t) = r(x(t))x(t)$$

When a biological population becomes too large, the per capita growth rate diminishes. This is because the individuals interfere with each other and are forced to compete for limited resources.

Verhulst introduce the carrying capacity of the environment constant, K , which represent the maximum size of the population that environment can sustain.

Thus, when $x(t) \rightarrow K$ then $r(x) \rightarrow 0$


when $x(t)$ is very small comparing with K , i.e. $x(t) \rightarrow 0$, then $r(x) \rightarrow r_0$, the competition for the resources does not influence the growth of the population, so this will grow as in Malthus model.

r_0 is called the unrestricted per capita growth rate.

Thus the function $r(x)$ has to interpolate the points $(0, r_0)$ and $(K, 0)$. Verhulst choose the linear interpolation, so

$$r(x) = r_0 \left(1 - \frac{x}{K} \right)$$

> `plot(2*(1-x/10),x=0..10);`



>

The logistic model is

$$\frac{d}{dt}x(t) = r x(t) \left(1 - \frac{x(t)}{K}\right)$$

$$x(0) = x_0$$

> `x0 := 'x0';`

$$x0 := x0$$

> `V_eq := diff(x(t), t) = r*x(t)*(1-x(t)/K);`

$$V_eq := \frac{d}{dt}x(t) = r x(t) \left(1 - \frac{x(t)}{K}\right)$$

> `ans := dsolve({V_eq, x(0)=x0}, x(t));`

$$ans := x(t) = \frac{K}{1 - \frac{e^{(-rt)}(-K + x0)}{x0}}$$


> `ans := simplify(ans);`

$$ans := x(t) = -\frac{x0 K}{-x0 - e^{(-rt)} K + e^{(-rt)} x0}$$


> `x_sol := unapply(rhs(ans), t, r, K, x0);`

$$x_sol := (t, r, K, x0) \rightarrow -\frac{x0 K}{-x0 - e^{(-rt)} K + e^{(-rt)} x0}$$

> `plot([x_sol(t, -0.5, 30, 30), x_sol(t, -0.5, 30, 28), x_sol(t, -0.5, 30, 10)], t=0..20, color=[black, red, blue]);`




```
> animate(x_sol(t,-0.5,30,30-x0),t=0..20,x0=0..30,frames=100);
```




```
>
```

```
plot([x_sol(t,0.5,30,50),x_sol(t,0.5,30,30),x_sol(t,0.5,30,10)],t=0..20,color=[red,black,blue]);
```



```
> animate(x_sol(t,0.5,30,50-x0),t=0..20,x0=0..50,frames=100);
```




>

Finding the logistic parameters: To apply the logistic model we need to find r and K . In this case we need to use two data in order to find them

```
> SUA_pop;
[3.92, 5.3, 7.23, 9.63, 12.86, 17.06, 23.19, 31.43, 39.81, 50.15, 62.94, 75.99, 91.97,
 105.71, 122.77, 131.66, 151.32, 179.32, 203.3, 226.54, 248.7]
```

```
> display(g1);
```



```
> x0:=SUA_pop[1];
```

$x0 := 3.92$

```
> x1:=SUA_pop[2];x2:=SUA_pop[11];
```

$x1 := 5.3$

$x2 := 62.94$

```
> syst:=x_sol(1,r,K,x0)=x1,x_sol(10,r,K,x0)=x2;
```

$$syst := -\frac{3.92 K}{-3.92 - e^{(-r)} K + 3.92 e^{(-r)}} = 5.3, -\frac{3.92 K}{-3.92 - e^{(-10 r)} K + 3.92 e^{(-10 r)}} = 62.94$$

```
> solve({syst},{r,K});
```



```

{K = 3.833656350 - 1.307700577 I, r = -0.1033047346 + 0.7765929325 I},
{K = 3.833656350 + 1.307700577 I, r = -0.1033047346 - 0.7765929325 I},
{r = -0.1546505963 + 1.461068823 I, K = 4.282961819 - 0.6883282258 I},
{K = 4.282961819 + 0.6883282258 I, r = -0.1546505963 - 1.461068823 I},
{K = 4.400679374 - 0.3525208398 I, r = -0.1810989743 + 2.135234725 I},
{r = -0.1810989743 - 2.135234725 I, K = 4.400679374 + 0.3525208398 I},
{r = -0.1928230331 + 2.806388449 I, K = 4.438313817 - 0.1097023212 I},
{K = 4.438313817 + 0.1097023212 I, r = -0.1928230331 - 2.806388449 I},
{K = 230.9709935, r = 0.3077116424 }

```

```


> r1 := .3077116424; K1 := 230.9709935;
      rI := 0.3077116424
      KI := 230.9709935

```

```

> g2 := plot(x_sol(t, r1, K1, x0), t=0..30, color=red);
> display(g1, g2);

```



>

Logistic parameters can sometimes be estimated by least squares. If the data values are separated by fixed time periods, h , then it is possible to remap the equations so least squares will work.

We have that

$$\frac{1}{x(t)} = \frac{1}{K} + \frac{K - x_0}{e^{(rt)} K x_0}$$

Suppose the data points are $(t_0, x_0), (t_1, x_1), (t_2, x_2), \dots, (t_n, x_n)$ with $t_{i+1} = t_i + h$, $i = 0, \dots, n-1$.

Then $t_{i+1} = t_i + h$ and the predicted value of x_{i+1} is given by

$$\frac{1}{x_{i+1}} = \frac{1}{K} + \frac{K - x_0}{e^{(r(t_i+h))} K x_0}$$

$$\frac{1}{x_{i+1}} = \frac{\frac{e^{(rh)} - 1}{K} + \frac{1}{K} + \frac{K - x_0}{e^{(rt_i)} K x_0}}{e^{(rh)}}$$

$$\frac{1}{x_{i+1}} = \frac{e^{(-rh)}}{x_i} + \frac{1 - e^{(-rh)}}{K}$$

Making $\frac{1}{x_i} = z_i$, $R = e^{(-rh)}$ and $Q = \frac{1 - R}{K}$ we get

$$z_{i+1} = R z_i + Q$$

For $i = 0$ we get that $Q = z_1 - R z_0$ thus

$$z_{i+1} = R z_i + z_1 - R z_0$$

The squared error is then given by

$$E(R) = \sum_{i=1}^n (z_i - (R(z_{i-1} - z_0) + z_1))^2$$

$E(R) \rightarrow \min$ then $E'(R) = 0$

$$2 \left(\sum_{i=1}^n (z_i - (R(z_{i-1} - z_0) + z_1)) (-z_{i-1} - z_0) \right) = 0$$

solving this equation with respect to R we obtain

$$R = \frac{\sum_{i=1}^n (z_i - z_1) (z_{i-1} - z_0)}{\sum_{i=1}^n (z_{i-1} - z_0)^2}$$

Returning to our notation we have that $r = -\frac{\ln(R)}{h}$ and $K = \frac{1 - R}{Q} = \frac{1 - R}{z_1 - R z_0}$

> **f := x -> 1/x;**

$$f := x \rightarrow \frac{1}{x}$$

> **z := map(f, SUA_pop);**

```

z := [ 0.2551020408, 0.1886792453, 0.1383125864, 0.1038421599, 0.07776049767,
       0.05861664713, 0.04312203536, 0.03181673560, 0.02511931675, 0.01994017946,
       0.01588814744, 0.01315962627, 0.01087311080, 0.009459842967,
       0.008145312373, 0.007595321282, 0.006608511763, 0.005576622797,
       0.004918839154, 0.004414231482, 0.004020908725 ]


> R:=sum((z[i+1]-z[2])*(z[i]-z[1]),i = 1 .. N-1)/sum((z[i]-z[1])^2,i
= 1 .. N-1);
                                     R := 0.7353475734

> r2:=-ln(R);
                                     r2 := 0.3074120026

> K2:=(1-R)/(z[2]-R*z[1]);
                                     K2 := 242.6715751

> g3:=plot(x_sol(t,r2,K2,x0),t=0..30,color=black):
> display(g1,g2,g3);

```



>

The logistic growth model with the treshold

Real populations are in danger of extinction if their size falls to a low level. Predation might eliminate the last few members completely, finding mates becomes more difficult, and lack of genetic diversity renders the population susceptible to epidemics. By constructing a per capita growth rate that is actually negative below some critical value T , there results a population model that tends to extinction if population size falls too low. Such a per capita growth rate is given as the right-hand side of the following modification of the logistic equation

$$\frac{d}{dt}x(t) = r x(t) \left(1 - \frac{x(t)}{K}\right) \left(\frac{x(t)}{T} - 1\right)$$

$$x(0) = x_0$$

where T is the threshold value and K is carrying capacity.

> LT_eq:=diff(x(t),t)=r*x(t)*(1-x(t)/K)*(x(t)/T-1);

$$LT_eq := \frac{d}{dt} x(t) = r x(t) \left(1 - \frac{x(t)}{K} \right) \left(\frac{x(t)}{T} - 1 \right)$$

> x0:='x0';

$$x0 := x0$$

> dsolve({LT_eq,x(0)=x0},x(t));

$$x(t) = e^{\text{RootOf}}$$

$$\ln(x0) T - \ln(-K + x0) T - \frac{Z T - \ln(e^{-Z} + K) T - r t T - K \ln(e^{-Z} + K - T) + \ln(e^{-Z} + K) K - \ln(x0) K + K \ln(x0 - T) + r t K}{+ K}$$

> dsolve(LT_eq,x(t),implicit);

$$t + \frac{K \ln(x(t) - T)}{r(T - K)} + \frac{\ln(x(t))}{r} - \frac{T \ln(-K + x(t))}{r(T - K)} + _CI = 0$$

> K:=100;T:=20;r:=0.5;

$$K := 100$$

$$T := 20$$

$$r := 0.5$$

> DEplot(LT_eq, x(t), t=0..30, [[x(0)=5], [x(0)=18], [x(0)=20], [x(0)=30], [x(0)=60], [x(0)=100], [x(0)=120]], stepsize=0.1, linecolor=blue);

